

# 1 Basic Rule Set

## 1.1 Configuration Objectives

```
/* -- configuration ----- */
    EvaluationPeriod 0.1
    RewardDelay 2
```

## 1.2 Set Definitions

```
/* -- app sets ----- */

DefineSet LocalUnhealthyApps :
App Local Where Health < 0, PeriodTune >= 0
Note "Local Apps with health value
below 0 (and not paused)"

DefineSet LocalUnhealthyTunableApps :
App Local Where Health < 0, PeriodTune < Limit
Note "Local Apps with health value
below 0 that still can be tuned"

DefineSet GlobalUnhealthyApps :
App Global Where Health < 0
Note "Global Apps with health value
below 0"

DefineSet LeastImportantTunableApps :
App Global Where LifeSignAge <= Limit,
NodeLifeSignAge <= Limit,
PeriodTune < Limit, Importance Min
Note "The least important apps in the system which
still can be tuned (so app and node have also
to be alive)"

DefineSet LeastImportantCompressableApps :
App Global Where CompressFactor < Limit,
```

## 1 Basic Rule Set

```
Importance Min
Note "The least important apps in the system which
still can be compressed"

DefineSet LeastImportantPausableApps :
App Global Where PeriodTune >= 0,
Importance Min
Note "The least important apps in the system
which still can be paused"

DefineSet MostImportantLocalCompressedHealthyApps :
App Local Where CompressFactor > 0, Health >= 0.6,
Importance Max
Note "The most important local apps that
are compressed"

DefineSet MostImportantLocalTunedHealthyApps :
App Local Where PeriodTune > 1, Health >= 0.6,
Importance Max
Note "The most important local apps that are tuned"

DefineSet MostImportantLocalPausedApps :
App Local Where PeriodTune < 0,
Importance Max
Note "The most important local apps that
are paused"

DefineSet MostImportantDeadApps :
App System Where LifeSignAge > Limit,
PeriodTune >= 0, Importance Max
Note "The most important system apps with expired
life signs (an which are not paused)"

DefineSet ForeignOrphans :
App LocalNonSystem Where SystemLifeSignAge > Limit
Note "Non-system apps running on our locale node
where we have lost contact to their systems"

/* -- node sets ----- */

DefineSet LocalUnhealthyNode :
Node Local Where Health <= 0.1
Note "Local node with health value of
0.1 and below"
```

```

DefineSet LocalHealthyNode :
Node Local Where Health >= 0.36
Note "Local node with health value of
0.36 and above"

DefineSet NonLocalSuitableNode :
Node NonLocal Where
Capacity % >= Demand + 0.15 (LocalUnhealthyApps),
LifeSignAge <= Limit, Health Max
Note "Non local alive node with enough
capacity (leave 15% free) to take an unhealthy app"

DefineSet
LocalNodeWithFifoAlsoAllowsPriorityScheduling :
Node Local Where Scheduling = 0, MaxScheduling > 0
Note "Local node which uses FIFO scheduling,
but also allows priority scheduling"

DefineSet
LocalNodeWithoutPreemptionAlsoAllowsPreemption :
Node Local Where Preemption = 0, MaxPreemption > 0
Note "Local node which uses no preemption, but
also allows preemption"

DefineSet AliveNodeWithLowestIdIsLocal :
Node System Where LifeSignAge <= Limit,
Id Min, HopCount = 0
Note "Contains the alive node with the lowest
id in the system if this is the local node"

DefineSet SuitableNodeForMostImportantDeadApps :
Node Global Where LifeSignAge <= Limit,
Capacity >= Demand (MostImportantDeadApps),
Health Max
Note "Global alive node with enough capacity
to take the most important dead app"

/* -- comm sets ----- */

DefineSet LocalUnhealthyComm :
Comm Local Where Health <= 0.15
Note "Local comm with health value of
0.15 and below"

```

```
DefineSet LocalHealthyComm :
Comm Local Where Health >= 0.5
Note "Local comm with health value of
0.5 and above"

DefineSet LocalVeryHealthyComm :
Comm Local Where Health >= 0.65
Note "Local comm with health value of
0.65 and above"

DefineSet
LocalCommWithFifoAlsoAllowsPriorityScheduling :
Comm Local Where Scheduling = 0, MaxScheduling > 0
Note "Local comm which uses FIFO scheduling,
but also allows priority scheduling"

DefineSet
LocalCommWithoutPreemptionAlsoAllowsPreemption :
Comm Local Where Preemption = 0, MaxPreemption > 0
Note "Local comm which uses no preemption, but
also allows preemption"
```

### 1.3 Condition Definitions

```
/* -- the conditions ----- */
DefineCondition RelocationMightBeUseful : C
Cardinal (LocalUnhealthyApps) > 0 And
Cardinal (LocalUnhealthyNode) > 0 And
Cardinal (LocalUnhealthyComm) = 0 And
Cardinal (NonLocalSuitableNode) > 0
Note "If there are unhealthy apps on an
unhealthy node, but comm is ok and there are
powerful nodes available, a relocation might
be useful"

DefineCondition CompressionMightBeUseful :
Cardinal (LocalUnhealthyApps) > 0 And
Cardinal (LocalUnhealthyComm) > 0 And
Cardinal (LeastImportantCompressableApps) > 0
Note "If there are unhealthy apps and an unhealthy
comm and there are still apps that can be
compressed, compression might be useful"
```

```

DefineCondition TuningMightBeUseful :
Cardinal (LocalUnhealthyApps) > 0 And
Cardinal (LocalUnhealthyComm) > 0 And
Cardinal (LeastImportantCompressableApps) = 0 And
Cardinal (LeastImportantTunableApps) > 0
Note "If there are unhealthy apps and an unhealthy
comm and there are no more apps that can be
compressed but still apps that can be
tuned, tuning might be useful"

DefineCondition TuningUnhealthyAppsMightBeUseful :
Cardinal (LocalUnhealthyTunableApps) > 0 And
Cardinal (LocalUnhealthyComm) = 0
Note "If there are unhealthy apps that still
can be tuned while comm is healthy, tuning
might be useful (also to make room for relocation)"

DefineCondition
TuningRemainingTunableAppsMightBeUseful :
Cardinal (LocalUnhealthyApps) > 0 And
Cardinal (LocalUnhealthyComm) = 0 And
Cardinal (LocalUnhealthyTunableApps) = 0 And
Cardinal (LeastImportantTunableApps) > 0
Note "If there are unhealthy apps while comm is
healthy and there are still tunable apps left,
tuning these apps might be useful
(also to make room for relocation)"

DefineCondition PausingMightBeUseful :
Cardinal (LocalUnhealthyApps) > 0 And
Cardinal (LeastImportantCompressableApps) = 0 And
Cardinal (LeastImportantTunableApps) = 0 And
Cardinal (LeastImportantPausableApps) > 0
Note "If there are unhealthy apps and there
are no more apps that can be compressed or
tuned but still apps that can be paused,
pausing might be useful
(also to make room for relocation) "

DefineCondition
PriorityNodeSchedulingMightBeUseful :
Cardinal (GlobalUnhealthyApps) > 0 And
Cardinal

```

```
(LocalNodeWithFifoAlsoAllowsPriorityScheduling) > 0
Note "If there are unhealthy apps in the system
and our local node is on fifo scheduling but allows
also priority based scheduling, a switch to
priority based scheduling might be useful"

DefineCondition
PriorityCommSchedulingMightBeUseful :
Cardinal (GlobalUnhealthyApps) > 0 And
Cardinal
(LocalCommWithFifoAlsoAllowsPriorityScheduling) > 0
Note "If there are unhealthy apps in the system and
our local comm is on fifo scheduling but allows
also priority based scheduling, a switch to
priority based scheduling might be useful"

DefineCondition NodePreemptionMightBeUseful :
Cardinal (GlobalUnhealthyApps) > 0 And
Cardinal
(LocalNodeWithoutPreemptionAlsoAllowsPreemption)>0
Note "If there are unhealthy apps in the system
and our local node allows preemption but preemption
is not active, a switch to preemption
might be useful"

DefineCondition CommPreemptionMightBeUseful :
Cardinal (GlobalUnhealthyApps) > 0 And
Cardinal
(LocalCommWithoutPreemptionAlsoAllowsPreemption)>0
Note "If there are unhealthy apps in the
system and our local comm allows preemption but
preemption is not active, a switch to preemption
might be useful"

DefineCondition UnpausingMightBePossible :
Cardinal (MostImportantLocalPausedApps) > 0 And
Cardinal (LocalHealthyNode) > 0 And
Cardinal (LocalHealthyComm) = MaxSet
Note "If there are local paused apps and the state
of the local node and comm is quite good,
unpausing might be possible"

DefineCondition UntuningMightBePossible :
Cardinal (MostImportantLocalTunedHealthyApps) > 0
```

```

And Cardinal (MostImportantLocalPausedApps) = 0
And Cardinal (LocalHealthyNode) > 0 And
Cardinal (LocalHealthyComm) = MaxSet
Note "If there are local tuned healthy apps and
no more local paused apps and the state of the
local node and comm is quite good ,
untuning might be possible"

DefineCondition UncompressingMightBePossible :
Cardinal
(MostImportantLocalCompressedHealthyApps) > 0 And
Cardinal (MostImportantLocalTunedHealthyApps) = 0
And Cardinal (MostImportantLocalPausedApps) = 0
And Cardinal (LocalHealthyNode) > 0 And
Cardinal (LocalVeryHealthyComm) = MaxSet
Note "If there are local compressed healthy apps
and no more local tuned or paused apps and the
state of the local node and comm is good ,
uncompresssing might be possible"

DefineCondition RestartMightBeUseful :
Cardinal (AliveNodeWithLowestIdIsLocal) > 0 And
Cardinal (MostImportantDeadApps) > 0 And
Cardinal (SuitableNodeForMostImportantDeadApps) > 0
Note "If the alive node with the lowest id is the
local node and this node detects important apps
with expired lifesigns and we have node capacity,
a restart of these apps might be a good idea
(so only one node does the restart, no conflicts)"

DefineCondition
TuningToEnableRestartMightBeUseful :
Cardinal (AliveNodeWithLowestIdIsLocal) > 0 And
Cardinal (MostImportantDeadApps) > 0 And
Cardinal (SuitableNodeForMostImportantDeadApps) = 0
And Cardinal (LeastImportantTunableApps) > 0
Note "If the alive node with the lowest id is the
local node and this node detects important apps
with expired lifesigns and we have no node capacity
but still tunable apps, tuning might be
useful to create capacity"

DefineCondition
PausingToEnableRestartMightBeUseful :

```

```
Cardinal (AliveNodeWithLowestIdIsLocal) > 0 And
Cardinal (MostImportantDeadApps) > 0 And
Cardinal (SuitableNodeForMostImportantDeadApps) = 0
And Cardinal (LeastImportantTunableApps) = 0 And
Cardinal (LeastImportantPausableApps) > 0
Note "If the alive node with the lowest id is
the local node and this node detects important apps
with expired lifesigns and we have no node capacity
and no tunable apps but still pausable apps,
pausing might be useful to create capacity"

DefineCondition ForeignOrphansExist :
Cardinal (ForeignOrphans) > 0
Note "Do foreign orphans exist once a
foreign CPS gets lost and we still
have remains running"
```

## 1.4 Action Definitions

```
/* -- the actions ----- */

DefineAction RelocateLocalUnhealthyApp :
Relocate LocalUnhealthyApps To NonLocalSuitableNode
Note "Relocate one local unhealthy
app to a suitable node"

DefineAction CompressLeastImportantApp :
Compress LeastImportantCompressableApps To
Current Limit
Note "Compress the least important
compressable app to its limit"

DefineAction TuneLeastImportantApp :
TunePeriodAndPriority LeastImportantTunableApps To
Current Limit
Note "Tune the least important tuneable
app to its limit"

DefineAction TuneLocalUnhealthyApps :
TunePeriodAndPriority LocalUnhealthyTunableApps To
Current Limit
Note "Tune a local unhealthy app to its limit"
```



```

DefineAction PauseLeastImportantApp :
TunePeriod LeastImportantPausableApps To -1
Note "Pause the least important pausable app"

DefineAction PauseLeastImportantAppFast :
TunePeriod LeastImportantPausableApps To -1
RewardDelay 0.3
Note "Pause the least important pausable app
(short reward delay to quickly make room)"

DefineAction ActivatePriorityNodeScheduling :
SetScheduling
LocalNodeWithFifoAlsoAllowsPriorityScheduling To 1
Adjust
Note "Activate priority based scheduling on local
node which currently is on fifo scheduling and
assign app priorities"

DefineAction ActivatePriorityCommScheduling :
SetScheduling
LocalCommWithFifoAlsoAllowsPriorityScheduling To 1
Note "Activate priority based scheduling on local
comms which are currently on fifo scheduling"

DefineAction ActivateNodePreemption :
SetPreemption
LocalNodeWithoutPreemptionAlsoAllowsPreemption To 2
Note "Activate preemption on local node
which currently does not use preemption"

DefineAction ActivateCommPreemption :
SetPreemption
LocalCommWithoutPreemptionAlsoAllowsPreemption To 2
Note "Activate preemption on local comms
which currently do not use preemption"

DefineAction UnpauseMostImportantLocalApp :
TunePeriodAndPriority MostImportantLocalPausedApps
To Current Limit
Note "Unpause the most important local app that
is paused"

DefineAction UntuneMostImportantLocalApp :
TunePeriodAndPriority

```

```
MostImportantLocalTunedHealthyApps To
Current -0.5 ExpLimitDown 1
Note "Untune the most important local app
that is tuned down by 0.5, don't go below 1"

DefineAction UncompressMostImportantLocalApp :
Compress MostImportantLocalCompressedHealthyApps To
Current -0.1 ExpLimitDown 0
Note "Uncompress the most important local app
that is compressed down by 0.1, don't go below 0"

DefineAction RestartDeadApp :
Restart MostImportantDeadApps On
SuitableNodeForMostImportantDeadApps
RewardDelay 0.3
Note "Restart the most important dead app
on a suitable alive node"

DefineAction KillForeignOrphans :
Remove All ForeignOrphans
Note "Kill all foreign orphans"
```

## 1.5 Rule Definitions

```
/* -- the rules ----- */

/* -- group 1: rules for crashes -- */

If RestartMightBeUseful Then RestartDeadApp
Note "Rule for restarting dead apps"

/* rules to make room for restarting crashed
apps in overload situations */

If TuningToEnableRestartMightBeUseful
Then TuneLeastImportantApp
Note "Rule to tune the least important app in case
there is no capacity to restart a crashed app"

If PausingToEnableRestartMightBeUseful
Then PauseLeastImportantAppFast
Note "Rule to pause the least important app in
case there is no capacity to restart a crashed"
```

```

app (this also includes crashed apps => the most
important apps will run)"

/* -- group 2: general rules for
overload situations -- */

If RelocationMightBeUseful
Then RelocateLocalUnhealthyApp
Note "Rule for relocation"

If CompressionMightBeUseful
Then CompressLeastImportantApp
Note "Rule for compression"

If TuningMightBeUseful Then TuneLeastImportantApp
Note "Rule for tuning"

If TuningUnhealthyAppsMightBeUseful
Then TuneLocalUnhealthyApps Lag 2
Note "Rule if we have unhealthy tunable
apps but comm is ok, lag due to race conditions"

If TuningRemainingTunableAppsMightBeUseful
Then TuneLeastImportantApp Lag 2
Note "Rule if we have unhealthy apps but comm
is ok and there are still tunable apps left,
lag due to race conditions"

If PausingMightBeUseful Then PauseLeastImportantApp
Note "Rule for pausing"

/* -- group 3: rules for scheduling issues -- */

If PriorityNodeSchedulingMightBeUseful
Then ActivatePriorityNodeScheduling
Note "Rule to fix scheduling issues on nodes"

If PriorityCommSchedulingMightBeUseful
Then ActivatePriorityCommScheduling
Note "Rule to fix scheduling issues on comms"

If NodePreemptionMightBeUseful
Then ActivateNodePreemption
Note "Rule to fix preemption issues on nodes"

```

## 1 Basic Rule Set

```
If CommPreemptionMightBeUseful
Then ActivateCommPreemption
Note "Rule to fix preemption issues on comms"

/* -- group 4: rules for undoing
overload measures -- */

If UnpausingMightBePossible
Then UnpauseMostImportantLocalApp
Note "Rule for undo pausing if the
situation gets better"

If UntuningMightBePossible
Then UntuneMostImportantLocalApp
Note "Rule for undo tuning if the
situation gets better"

If UncompressingMightBePossible
Then UncompressMostImportantLocalApp
Note "Rule for undo compression if the
situation gets better"

/* -- group 5: rules to kill foreign orphans -- */

If ForeignOrphansExist Then KillForeignOrphans
Note "Rule for killing foreign orphans once a
foreign CPS gets lost and we still have
remains running"
```