

1 Contextfree LL(1) - grammar – *Rango*

```
RuleContainer = Configuration RuleSet.

Configuration = {ConfigurationDirective}.

ConfigurationDirective =
    ('EvaluationPeriod' | 'RewardDelay' |
     'LearningRate' | 'InterferenceWeight' |
     'MinLearningRateScale' | 'FitnessWeightAppHP' |
     'FitnessWeightAppImportance' | 'FitnessWeightApp' |
     'FitnessWeightNode' | 'FitnessWeightComm' |
     'FitnessLifeSignScale' | 'FitnessCrashHealth')
    Real |
    'FitnessSicknessBoost' Real Real |
    'FitnessScope' BasicScope | 'FitnessMode' Integer |
    'SkipImpossibleActions' ('On' | 'Off') | Lag.

RuleSet = {Rule | NamedSet | NamedCondition | NamedAction}.

Rule = 'If' Condition
      'Then' Action [Lag] [Reward] [Experience] [Note].
```

1.1 Condition production rules

```
Condition = ConditionDefinition | ConditionName.

NamedCondition = 'DefineCondition' ConditionName ':'
                ConditionDefintion [Note].

ConditonName = Identifier.

ConditionDefinition = SubCondition {'And' SubCondition}.

SubCondition = CardinalCondition | AverageCondition |
               MinCondition | MaxCondition.
```

```
CardinalCondition = 'Cardinal' '(' Set ')'
                  DOperator (Integer|'MaxSet' [Real]).

AverageCondition = 'Average' '('
                  AttributeAndSet DOperator Real.

MinCondition = 'MinElement' '('
              AttributeAndSet DOperator Real.

MaxCondition = 'MaxElement' '('
              AttributeAndSet DOperator Real.

Set = AppSet | NodeSet | CommSet.

AttributeAndSet = AppAttributeAndSet |
                 NodeAttributeAndSet |
                 CommAttributeAndSet.

AppAttributeAndSet = AppSet ')' AppAttribute.

NodeAttributeAndSet = NodeSet ')' NodeAttribute.

CommAttributeAndSet = CommSet ')' CommAttribute.
```

1.2 Action production rules:

```
Action = (ActionDefinition | ActionName).

NamedAction = 'DefineAction' ActionName ':'
             ActionDefinition [Note].

ActionName = Identifier.

ActionDefinition = ActionType ['RewardDelay' Real].

ActionType = StartApp | StopApp | RemoveApp | RelocateApp |
             RestartApp | TuneApp | CompressApp |
             SmoothPeriodExtensionApp | SchedulingMode |
             PreemptionMode | SetPriority | MonitorPeriod |
             DoNothing.

StartApp = 'Start' [SetOption] AppSet.
```

```

StopApp = 'Stop' [SetOption] AppSet.

RemoveApp = 'Remove' [SetOption] AppSet.

RelocateApp = 'Relocate' [SetOption] AppSet
              'To' [SetOption] NodeSet ['ResetTuning'].

RestartApp = 'Restart' [SetOption] AppSet
             'On' [SetOption] NodeSet ['ResetTuning'].

TuneApp = ('TunePeriod' | 'TunePriority' | 'TuneDeadline' |
           'TuneReactionTime' | 'TunePeriodAndPriority' |
           'TunePeriodAndDeadline' |
           'TunePeriodAndReactionTime') [SetOption]
AppSet 'To' Factor.

CompressApp = 'Compress' [SetOption]
AppSet 'To' Factor.

SmoothPeriodExtensionApp = 'Extend' [SetOption]
AppSet 'To' ('On' | 'Off').

SchedulingMode = 'SetScheduling' [SetOption]
(NodeSet 'To' Integer ['Adjust'] |
CommSet 'To' Integer).

PreemptionMode = 'SetPreemption' [SetOption]
(NodeSet | CommSet) 'To' Integer.

SetPriority = 'SetPriority' [SetOption]
(AppSet) 'To' Factor.

MonitorPeriod = 'SetMonitoringPeriod' [SetOption]
(AppSet | NodeSet | CommSet) 'To' Factor.

DoNothing = 'DoNothing'

Reward = 'Reward' Real.

Experience = 'Experience' Integer.

Lag = 'Lag' Integer.

Note = 'Note' String.

```

1.3 Set production rules

```
AppSet = (AppSetDefinition | AppSetName).

NodeSet = (NodeSetDefinition | NodeSetName).

CommSet = (CommSetDefinition | CommSetName).

NamedSet = 'DefineSet'
           (AppSetName ':' AppSetDefinition) |
           (NodeSetName ':' NodeSetDefinition) |
           (CommSetName ':' CommSetDefinition) [Note].

AppSetName = Identifier.

NodeSetName = Identifier.

CommSetName = Identifier.

AppSetDefinition = 'App' (DirectAppId |
                        Scope [AppQueryClausesList]).

NodeSetDefinition = 'Node' (DirectId |
                            Scope [NodeQueryClausesList]).

CommSetDefinition = 'Comm' (DirectId |
                            Scope [CommQueryClausesList]).
```

1.4 Support production rules

```
SetOption = 'All'.

Scope = BasicScope | CompositeScope.

BasicScope = 'Global' | 'System' | 'Local'.

CompositeScope = 'NonLocal' | 'NonSystem' |
                 'LocalSystem' | 'LocalNonSystem' |
                 'NonLocalSystem' | 'NonLocalNonSystem'.

DirectAppId = DirectId ['Started' | 'NonStarted']
```

```

DirectId = String ['Local'].

Unsigned = Digit {Digit}.

Integer = ['-'] Unsigned.

Real = Integer [',', Unsigned].

Identifier = Letter {Letter | Digit}.

Digit = '0' | '1' | '2' | '3' | '4' | '5' | '6' | '7' |
        '8' | '9'.

Letter = ('a' .. 'z') | ('A' .. 'Z') | '_' | '#'.

String = '"' {Letter | Digit | '/' | '<' | '>' |
              '!' | '?' | '=' | '(' | ') | '*' | '+' |
              '-' | ';' | ':' | '.' | ',' | '}' '"'}.

Factor = AbsoluteValue | RelativeValueToCurrent.

AbsoluteValue = Real.

RelativeValueToCurrent =
    'Current' LinearExpression [('Limit' [Real] |
    ('ExpLimitUp' | 'ExpLimitDown') Real]].

LinearExpression = ['*' Real] [('+' | '-') Real].

RelativeValueToDemand = 'Demand' LinearExpression
    '(' [SetOption] AppSet ')'.

AppQueryClausesList = 'Where' AppClauses {'', AppClauses}.

AppClauses = AppAttribute (DOperator AppValue | MOperator).

AppAttribute = 'Importance' | 'NodeDemand' [%'] |
    'CommDemand' [%'] | 'PeriodTune' |
    'MaxPeriodTune' | 'PriorityTune' |
    'DeadlineTune' | 'ReactionTimeTune' |
    'CompressFactor' | 'MaxCompressFactor' |
    'Health' | 'NodeHealth' | 'CommHealth' |
    'HopCount' | 'AvgCommTime' |
    'LastCommTime' | 'Period' | 'Priority' |

```

```

        'Deadline' | 'ReactionTime' |
        'PeriodExtension' | 'MonitoringPeriod' |
        'Id' | 'LifeSignAge' |
        'NodeLifeSignAge' | 'SystemLifeSignAge'.

NodeQueryClausesList =
    'Where' NodeClauses {' ',' NodeClauses}.

NodeClauses = NodeAttribute
    (DOperator NodeValue | MOperator).

NodeAttribute = ('Capacity' | 'NodeCapacity' |
    'CommCapacityBest' | 'CommCapacityWorst' |
    'CommInCapacityBest' |
    'CommInCapacityWorst' |
    'CommOutCapacityBest' |
    'CommOutCapacityWorst') ['%'] |
    'Health' | 'CommHealth' | 'HopCount' |
    'AvgCommTime' | 'LastCommTime' |
    'Scheduling' | 'Preemption' |
    'MaxScheduling' | 'MaxPreemption' |
    'MonitoringPeriod' | 'Id' | 'LifeSignAge' |
    'NodeLifeSignAge' | 'SystemLifeSignAge'.

CommQueryClausesList =
    'Where' CommClauses {' ',' CommClauses}.

CommClauses = CommAttribute
    (DOperator CommValue | MOperator).

CommAttribute = 'Capacity' ['%'] |
    'InCapacity' ['%'] | 'OutCapacity' ['%'] |
    'Health' | 'Scheduling' | 'Preemption' |
    'MaxScheduling' | 'MaxPreemption'.

DOperator = '<' | '>' | '=' | '!=' | '<=' | '>='.

MOperator = 'max' | 'min'.

AppValue = AbsoluteValue | RelativeValueToLimit.

RelativeValueToLimit = 'Limit' LinearExpression.

NodeValue = AbsoluteValue | RelativeValueToDemand |

```

1.4 Support production rules

```
RelativeValueToLimit.
```

```
CommValue = AbsoluteValue | RelativeValueToDemand.
```